

# Incremental Computing in Haskell

## CS240H Project Report

Jiyue WANG and Kaixi RUAN

March 18, 2016

## 1 Introduction

In this project, we implement a Haskell library for incremental computations. Incremental computing, or self-adjusting computing, is the idea of tracking data and control dependencies in order to selectively re-evaluate only parts of the computations given some changed inputs.

A simple and well-known example of incremental computing is the spreadsheet. In a spreadsheet, a cell contains either some simple data, or an equation describing how the result should be derived from values in other cells. It is critical for a spreadsheet to avoid re-evaluating all the cells when only one cell has changed value.

Jane Street recently released their open-source library written in OCaml named *Incremental*[5], which concretizes this idea. It is natural for us to inquire whether a similar library exists for Haskell.

It turns out that there are already some existing libraries in Haskell related to incremental computing: *Adaptive*[3] is one directly related to Incremental Computing, but the library was based on some theoretical results in the year 2002[1] and the theory has evolved since then. There is also a Functional Reactive Programming library in Haskell called *Reactive-banana*[2]. However, this library is mostly focused on GUI applications rather than computations. Thus, we decided to write a library which provides similar functionality as Incremental by Jane Street.

In the following sections, we describe how a library like Jane Street's Incremental is implemented using Haskell.

## 2 Incremental in a nutshell

### 2.1 Incremental DAG

The key idea behind Incremental Computing is a Directed Acyclic Graph(DAG).

In an Incremental DAG, a node represents an incremental value in the computation: it can be a simple variable input, or it can depend on some other incremental values.

An edge exists from node  $n$  to node  $m$  if the value of node  $m$  depends on the value of node  $n$ . We say that node  $m$  is a parent of node  $n$ .

It is possible that the user creates a cyclic graph by mistake. The library will then detect and throw exception during stabilization.

### 2.2 Demo

The following example shows how to use out current Incremental library.

Listing 1: Example

```
demo :: StateIO ()
demo = do
  -- (x + y) + (y * z)
  x <- var (1 :: Int)
  y <- var (2 :: Int)
  z <- var (3 :: Int)
  t1 <- map2 (+) (watch x) (watch y)
  t2 <- map2 (*) (watch y) (watch z)
  w <- map2 (+) t1 t2
  o <- observe w

  stabilize
  printObs o

  writeVar y 4
  stabilize
  printObs o
```

Listing 2: Output of demo

```
-- New Var created [NodeRef ID = 1]
-- New Var created [NodeRef ID = 2]
-- New Var created [NodeRef ID = 3]
--State.createObserver Observer ID = 7
--...
-- State.becameNecessary [NodeRef ID = 3]
-- State.unlinkDisallowedObs begin
```

```

-- Current root set is [[PackedNode]
  =>[NodeRef ID = 1],[PackedNode]
  =>[NodeRef ID = 2],[PackedNode]
  =>[NodeRef ID = 3],[PackedNode]
  =>[NodeRef ID = 4],[PackedNode]
  =>[NodeRef ID = 5],[PackedNode]
  =>[NodeRef ID = 6]]
-- State.recompute node [NodeRef ID = 3]
-- State.recompute node [NodeRef ID = 2]
-- State.recompute node [NodeRef ID = 5]
-- State.recompute node [NodeRef ID = 1]
-- State.recompute node [NodeRef ID = 4]
-- State.recompute node [NodeRef ID = 6]
[Observer ID = 7] = 9
-- State.unlinkDisallowedObs begin
-- Current root set is [[PackedNode]
  =>[NodeRef ID = 2]]
-- State.recompute node [NodeRef ID = 2]
-- State.recompute node [NodeRef ID = 5]
-- State.recompute node [NodeRef ID = 4]
-- State.recompute node [NodeRef ID = 6]
[Observer ID = 7] = 17

```

In the previous example, a simple computation  $w = (x + y) + (y * z)$  is created. When the value of one input ( $y$ ) changes, only the nodes that depend on  $y$  are re-evaluated. Here we provide a brief explanation for different functionality of the library:

**Variable** Users can create variables through `var`, and they can later change the value of a variable by calling `writeVar`.

**Observer** Users add observer to read the value of the node in interest. For instance, we declare `w` as an observer in the example code where its value can be printed out using `printObs`.

**Necessary Node** A node is marked necessary if there is a path from that node to some observed node. An observed node is automatically necessary.

**Operation** mainly describes how an incremental is defined/related to other incrementals. Simple examples of operations are `map`, `map2` (used in demo). Like variables, they are also represented by `Node`, but of different `Kind`.

**Stabilize** After building the graph/change the value, use `stabilize` to recompute the DAG. Only the part of the graph that depends on the actual change would be re-evaluated. This is the most important call in Incremental. Reading the value of an observer without stabilization will throw exceptions.

### 3 Implementation

Representing a dynamic graph in Haskell is not as straightforward as a tree. For efficiency consideration, we decided to give up purity and use `IORef`.

The good news is that we only need to keep one copy of each node, while on the other hand, most of the manipulation will live in IO monad.

#### 3.1 Node

`Node` represents a node in the DAG. Each node needs to maintain a list of fields which might be updated during the stabilization. To avoid copying large record, we organize the fields in a hierarchical structure and use `Lens.Simple` to get access to a specific field efficiently. Listing[3] gives code snippets related to `Node`.

Listing 3: Node

---

```

data Node a = Node {
  _kind      :: Kind a
  , _value   :: ValueInfo a
  , _edges   :: Edges
}

data Kind a =
  forall b. Eq b => ArrayFold ...
| forall b. Eq b => Bind {
  func  :: b -> StateIO (NodeRef a)
  , lhs  :: NodeRef b
  , rhs  :: Maybe (NodeRef a)
  , nodesCreatedInScope :: [PackedNode]
}
| Const a
| Variable {
  mvalue :: a
  , setAt  :: StabilizationNum
  , valueSetDuringStb :: !(Maybe a)
}
| forall b. Eq b => Map (b -> a) (NodeRef b)
...

data Edges = Edges {
  _parents :: Set PackedNode
  , _obsOnNode :: Set ObsID
}

data Scope = Top
  | forall a. Eq a => Bound (NodeRef a)

data NodeRef a = Ref (IORef (Node a))
  !Unique -- node id
  !Scope  -- scope created in

data PackedNode = forall a. Eq a =>
  PackedNode (NodeRef a)

```

---

`kind` could be `Variable`, `Map`, `Bind`, etc., which represents the type of the node. It also stores references (parent-to-child edge) to all possible children when it is first created. However, the child node does not necessarily has an edge to its parents. The child-to-parent edge is added (from parent) only when the parent becomes

necessary and it is removed once the parent is unnecessary.

**value** not only contains the current node value but also contains extra information to help decide whether the value is stale.

**edges** stores the topological information of the graph. It contains references to parent nodes(child-to-parent edge) as well as observers watching the current node.

**Unique** gives a unique identifier for each `NodeRef` that helps to compare nodes of different types without dereferencing the `IRef (Node a)`.

**Scope** indicates the scope in which the node is created. A user could introduce a new node in the ‘global’ scope (`Top`), or on the RHS of a `Bind` node. This is useful when the recomputation involves a `Bind` node. As both **id** and **scope** is immutable during the lifetime of a node, we could keep them outside `Node a`, thus saving one layer of indirectness.

**PackedNode** is a convenience wrapper over nodes of different types. This allows us to store heterogeneous parent/child nodes.

### 3.2 Observer

Users can only change the value of a `Variable` node, but they can read other kind of node by adding an `Observer` to the node in interest. An `InUse` observer makes the observed node necessary. Remember that only necessary nodes will appear in the DAG and update during stabilization.

Listing 4: Observer

---

```

newtype Observer a = Obs ObsID

type ObsID = Unique

data InterObserver a = InterObs {
  _obsID :: !ObsID
  , _state :: !ObsState
  , _observing :: !(NodeRef a)
}

data ObsState = Created
              | InUse
              | Disallowed
              | Unlinked

data PackedObs = forall a. Eq a =>
                PackObs (InterObserver a)

```

---

### 3.3 State

We need an environment like `State` monad to keep track of the DAG and observers. To incorporate

`IO` monad as well, we use the monad transformer `StateT` to stack them into a new monad `StateIO`.

Listing 5: State

---

```

type StateIO a = StateT StateInfo IO a

data StateInfo = StateInfo {
  _info      :: StatusInfo
  , _recHeap  :: Set PackedNode
  , _observer  :: ObserverInfo
  , _varSetDuringStb :: [PackedVar]
}

```

---

**info** keeps track of status related information, including whether the program is during a stabilization, the stabilization number, current scope and debug information, etc.

**recHeap** is a somewhat misnamed field. It is used be a minimum heap which stores the nodes that needs to be recomputed during next stabilization. Later, we use DFS-based topological sorting to update nodes and this field becomes a set of root nodes for DFS.

**observer** is a map of observer ID to instances. Currently, we use a standard map which based on size balanced binary trees. It could be easily replaced by other containers like `IntMap` to improve performance.

**varSetDuringStb** is a list of variables set during stabilization, used in asynchronous stabilization.

### 3.4 Stabilization

After the user adds observers or make changes to variables, they need to call `stabilize` or `stabilizeAsync` to trigger the recomputation. The algorithm is a little complicated because of the `Bind` node.

#### 3.4.1 Bind node

Listing 6: Bind node

---

```

data Kind a = ...
| forall b. Eq b => Bind {
  func :: b -> StateIO (NodeRef a)
  , lhs  :: NodeRef b
  , rhs  :: Maybe (NodeRef a)
  , nodesCreatedInScope :: [PackedNode]
}
...

```

---

For a static graph, the algorithm is straightforward. First, it starts DFS from the nodes in `recHeap` and gets a list of nodes. It then recomputes nodes in the list sequentially and updates all the necessary nodes.

This algorithm will not work with a `Bind` node (see Listing[6]) that generates the graph on the fly. To deal with `Bind` node, we do the following modification (see Algorithm[1]). Note that the ‘else’ part actually solve three possible cases and we check the stabilization number before recomputing the node to avoid duplicated work.

- a. `rhs` is `Nothing`.
- b. Only `lhs` changes.
- c. Both `lhs` and `rhs` changes.

```

if stbNum(LHS) < stbNum(Current) then
  (LHS does not change);
  copy the value of RHS node;
else
  (Regenerate RHS);
  run func;
  recompute RHS nodes recursively;
  update rhs if necessary;
end

```

**Algorithm 1:** Recompute `Bind` node

### 3.4.2 Asynchronous Stabilization

Considering that a large graph may take time to recompute, we provide `stabilizeAsync`, `waitForStb` and a helper function `amStabilizing` to forward the recomputation to another thread. During the stabilization, the user is allowed to create and modify nodes as well as observers. However, all the actions taking place during current stabilization will not take effect until the next stabilization. Check out Listing[7] for an example.

Listing 7: Asynchronous Stabilization

---

```

exampleAsync :: StateIO ()
exampleAsync = do
  v1 <- var (5 :: Int)
  b1 <- (const True) >>=|
    (\_ -> expensiveWork 1000 >> return
      (watch v1))

  ob <- observe b1
  stabilizeAsync

  ob' <- observe b1

  waitForStb -- ob' is still not valid
  printObs ob -- ob is valid

```

```

stabilize
printObs ob' -- ob' is valid now

expensiveWork :: Int -> StateIO ()
expensiveWork n = lift (putStr $ (take n $
  repeat '.'))

```

---

This is also a counterexample showing that it is user’s responsibility to make sure the function fed into `Bind` node have no other side effect except for introducing new nodes or observers. It is easy to crash the system by lifting inappropriate IO actions and thus we are trying to restrict misuse on the type level to improve safety.

## 3.5 Exception Handling

Runtime exception are inevitables as the graph is changing dynamically. However, the exception handling functions in `Control.Exception.Base` only work with IO monad, not monad transformers. Exception handling in monad transformers could be tricky (see [4]). Fortunately, the `exceptions` package gives a clean solution in our case.

## 3.6 Garbage Collection

Generally speaking, nodes created in `Top` scope should never be garbage collected because they are static and might be observer by the user at some time. However, nodes created on RHS of `Bind` could be GC when it is no longer necessary. This is done by unlinking the unnecessary node from the graph and the Haskell runtime system will do GC for us.

# 4 Future Work

## 4.1 Testing

Currently, we don’t have enough time to write thorough test cases for the library. As most of the function has `StateIO` in their signature, we plan to use `Test.QuickCheck.Monad` help us.

## 4.2 Complete the Functionality

So far, our library provides sufficient elementary functions like `map`, `map2`, `bind`, etc. To make the user’s life easier, we are adding more functionality like `arrayFold`, `ifElseThen`, `sum`, etc. In addition, it will be nice to log the current state of graph in local file system and recover/resume the computation. Also, we are completing the exception handling mechanism to help reduce runtime exception caused by misuse.

### 4.3 Improve the Algorithm

Our current algorithm maintains a global state to keep track of the graph and uses a DFS-based topological sorting algorithm to do recomputation. A tempting future work is to make the recomputation parallel, which is not possible using current algorithm.

## References

- [1] U. A. Acar, G. E. Blelloch, and R. Harper. Adaptive functional programming. *ACM Transactions on Programming Languages and Systems (TOPLAS)*, 28(6):990–1034, 2006.
- [2] H. Apfeldmus. Reactive-banana. <https://wiki.haskell.org/Reactive-banana>. 2016-01-04.
- [3] M. Carlsson. Adaptive. <https://hackage.haskell.org/package/Adaptive>. 2013-01-28.
- [4] M. Snoyman. Exceptions and monad transformers. <https://www.schoolofhaskell.com/user/snoyberg/general-haskell/exceptions/exceptions-and-monad-transformers>. 2013-08-22.
- [5] J. Street. incremental. [https://github.com/janestreet/incremental/blob/master/src/incremental\\_intf.ml](https://github.com/janestreet/incremental/blob/master/src/incremental_intf.ml). 2015-08-17.